# How to draw a tetrahedron with angle marked?

How to draw a tetrahedron with this point of view and with specific angle marked?

Thank you

I just find a sample code but I have no idea how to edit it.

\psset{viewpoint=40 10 10 rtp2xyz,lightsrc=viewpoint,Decran=30}
\begin{pspicture}[solidmemory](-2,-2)(3,3)
\psSolid[object=tetrahedron,r=3,action=draw*,name=T,num=1 2 3]% without 0
\psSolid[object=point,definition=solidgetsommet,args=T 0,text=Top,pos=uc]% Point T0 (top)
\end{pspicture}

• I have found some samples but I have no idea how to edit it. – Alsace lei Jun 7 '18 at 8:48
• \psset{viewpoint=40 10 10 rtp2xyz,lightsrc=viewpoint,Decran=30} \begin{pspicture}[solidmemory](-2,-2)(3,3) \psSolid[object=tetrahedron,r=3,action=draw*,name=T,num=1 2 3]% without 0 \psSolid[object=point,definition=solidgetsommet,args=T 0,text=Top,pos=uc]% Point T0 (top) \end{pspicture} – Alsace lei Jun 7 '18 at 8:48
• Add as an edit to you question. Don't comment a lot of code like that. – gvgramazio Jun 7 '18 at 8:53
• Are you new to LaTeX or, more specifically, to packages to draw in LaTeX like TikZ, pgfplots, PSTricks, etc. ? – gvgramazio Jun 7 '18 at 8:55
• I am new to use LaTeX to draw graphics – Alsace lei Jun 7 '18 at 9:11

Here's a version of your drawing in Metapost, which you might like to learn.

Here's a lualatex program to create that as a standalone graphic.

\documentclass[border=5mm]{standalone}
\usepackage{luatex85}
\usepackage{luamplib}
\begin{document}
\mplibtextextlabel{enable}
\begin{mplibcode}
beginfig(1);
z0 = origin;
z1 = 200 right rotated -20;
z2 = 240 right rotated 10;
z3 = 210 right rotated 60;

path a[];
a1 = .85[z1,z3] {dir  20} .. .83[z2,z3];
a2 = .90[z2,z3] {dir 210} .. .90[z0,z3];
a3 = .86[z0,z3] {dir -30} .. .88[z1,z3];

draw a1 withcolor 3/4 blue;
fill a2 -- z3 -- cycle withcolor .9[blue, white];
draw a2 withcolor .7[blue, white];
draw a3 withcolor 3/4 blue;

draw z0--z2 dashed withdots scaled 1/2;
draw z0--z1--z2--z3--cycle;
draw z1--z3;

drawoptions(withcolor 3/4 blue);
label.lrt(textext("$\alpha$") rotated 40, point 1/2 of a1);
label.lft("$\beta$", z3 shifted 20 left);
interim ahangle := 20;
drawarrow z3 shifted 20 left {right}..z3 shifted -(4,5);
label.bot("$\gamma$", point 1/2 of a3);

drawoptions();

label.llft("$A$", z0);
label.lrt("$B$", z1);
label.rt("$C$", z2);
label.top("$V$", z3);

endfig;
\end{mplibcode}
\end{document}


Note that there's no native 3D support in MP. If you want to do lots of 3D work, then Asymptote is probably the best of the TeX-friendly tools. But there are 3D libraries for PSTricks and for TikZ if you are keen to try them. There are even also 3D libraries for Metapost (like this one) but none of them is very easy to use.